SingularValueDecomposition

`SingularValueDecomposition[`*m*`]`
gives the singular value decomposition for a numerical matrix *m*, as a list of matrices {*u*, *w*, *v*}, where *w* is a diagonal matrix, and *m* can be written as *u*.*w*.`Conjugate[Transpose[`*v*`]]`.

`SingularValueDecomposition[{`*m*`, `*a*`}]`
gives the generalized singular value decomposition of *m* with respect to *a*.

`SingularValueDecomposition[`*m*`, `*k*`]`
gives the singular value decomposition associated with the *k* largest singular values of *m*.

`SingularValueDecomposition[{`*m*`, `*a*`}, `*k*`]`
gives the generalized singular value decomposition associated with the *k* largest singular values.

MORE INFORMATION

EXAMPLES

Basic Examples (1)

In[1]:= `{u, w, v} = SingularValueDecomposition[{{1, 2}, {1, 2}}]`

Out[1]= $\left\{\left\{\left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}\right\}, \left\{\left\{\sqrt{10}, 0\right\}, \{0, 0\}\right\}, \left\{\left\{\frac{1}{\sqrt{5}}, -\frac{2}{\sqrt{5}}\right\}, \left\{\frac{2}{\sqrt{5}}, \frac{1}{\sqrt{5}}\right\}\right\}\right\}$

In[2]:= `u.w.Transpose[v]`

Out[2]= `{{1, 2}, {1, 2}}`

Scope (3)

Generalizations & Extensions (2)

Options (1)

Applications (2)

m is a 2×2 matrix:

In[1]:= `m = {{1, -1}, {1, 2}};`

Find its singular value decomposition:

In[2]:= `{u, w, v} = SingularValueDecomposition[N[m]]`

Out[2]= `{{{-0.289784, 0.957092}, {0.957092, 0.289784}}, {{2.30278, 0.}, {0., 1.30278}}, {{0.289784, 0.957092}, {0.9`

©1988–2008 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com

The columns of *v* give the directions of minimal and maximal stretching of vectors by *m.x*:

In[3]:=
```
ParametricPlot[m.{Cos[t], Sin[t]}, {t, 0, 2 Pi}, Epilog →
  {Thickness[0.01], {Red, Line[{{0, 0}, m.v[[All, 1]]}]}, {Green, Line[{{0, 0}, m.v[[All, 2]]}]}}]
```

Out[3]=



The columns of *u* give the directions of minimal and maximal stretching of vectors by *x.m*:

In[4]:=
```
ParametricPlot[{Cos[t], Sin[t]}.m, {t, 0, 2 Pi}, Epilog →
  {Thickness[0.01], {Red, Line[{{0, 0}, u[[All, 1]].m}]}, {Green, Line[{{0, 0}, u[[All, 2]].m}]}}]
```

Out[4]=



Here is some randomly generated data:

In[1]:=
```
t = RandomReal[{0, 2 Pi}, 20];
y = Sin[t] + .5 Cos[t] + RandomReal[.1, 20];
data = Transpose[{t, y}];
ListPlot[data]
```

Out[1]=



Construct a design matrix for fitting the data to basis functions $\{1, \sin(t), \cos(t)\}$:

In[2]:=
```
m = Map[Function[{s}, {1., Sin[s], Cos[s]}], t]
```

Out[2]=
```
{{1., -0.207075, 0.978325}, {1., 0.850917, -0.5253}, {1., -0.667871, 0.744277}, {1., -0.977369, -0.211543},
 {1., 0.606033, -0.795439}, {1., 0.187669, -0.982232}, {1., 0.155239, -0.987877}, {1., 0.467998, 0.88373},
 {1., -0.348298, -0.937384}, {1., -0.218383, -0.975863}, {1., -0.886513, -0.462705}, {1., 0.708333, -0.7058
```

©1988–2008 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com

Find the condensed singular value decomposition:

```
In[3]:= {u, w, v} = SingularValueDecomposition[m, 3]

Out[3]= {{{-0.168775, 0.337533, -0.0535807}, {-0.242391, -0.0974851, 0.305601}, {-0.17958, 0.279742, -0.228151}, {
   {-0.232561, -0.023254, -0.348334}, {-0.255162, -0.16874, 0.209855}, {-0.263732, -0.214059, 0.0518847}, {
   {-0.168087, 0.3389, 0.0443789}, {-0.191939, 0.210494, -0.309924}, {-0.240308, -0.0857236, 0.315888}, {-0
   {-0.237555, -0.0525118, -0.333336}, {-0.25095, -0.145451, 0.249378}, {-0.169793, 0.327127, 0.116086}, {-
  {{4.51497, 0., 0.}, {0., 3.50329, 0.}, {0., 0., 2.7096}}, {{-0.975999, 0.217647, 0.00749802}, {-0.00472291
```

Find a vector x that minimizes $\|m.x - y\|_2$:

```
In[4]:= x = v.((Transpose[u].y) / Diagonal[w])

Out[4]= {0.0446112, 0.985759, 0.49371}
```

The components of x are the coefficients given by `Fit`:

```
In[5]:= Fit[data, {1, Sin[s], Cos[s]}, s]

Out[5]= 0.0446112 + 0.49371 Cos[s] + 0.985759 Sin[s]
```

```
In[6]:= Show[ListPlot[data], Plot[%, {s, 0, 2 Pi}]]
```

Out[6]=



Properties & Relations (4)

Possible Issues (1)

SEE ALSO

SingularValueList ▪ **Norm** ▪ **PseudoInverse** ▪ **LeastSquares** ▪ **QRDecomposition** ▪ **SchurDecomposition**

TUTORIALS

- Advanced Matrix Operations

RELATED LINKS

- Demonstrations with SingularValueDecomposition (Wolfram Demonstrations Project)
- Implementation notes: Numerical and Related Functions

MORE ABOUT

- Matrices and Linear Algebra
- Matrix-Based Minimization
- Matrix Decompositions
- Signal Processing

©1988–2008 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com